

# BIOMETRICS INFORMATION

(You're 95% likely to need this information)


PAMPHLET NO. # 60

DATE: September 16, 1999

SUBJECT: MANOVA: Profile Analysis – an example using SAS®

A *profile* can be thought of as a broken line that graphically joins interdependent observations which are measured on the same experimental unit. Observations could be inter-dependent because they were taken over distinct time periods or observed from different competing taxa. An average profile consists of the average of the observations within each treatment level, as in Figure 1 below. Profile analysis compares these average profiles from two or more treatment groups<sup>1</sup>.

Let's begin by considering an example experiment. A researcher is studying the long-term effects of three brushing treatments (a herbicide, a manual brushing, and an untreated control) on the density of various regenerated tree species<sup>2</sup>. Each treatment is randomly applied to 5 square plots that are  $\frac{1}{4}$  ha in size. Within each plot, the researcher also establishes 3 circular subplots to measure the density of four different naturally regenerated tree species (fir, larch, pine and birch), several years following the application of the brushing treatments. Figure 1 shows the average profiles for the three treatment groups using the data in Appendix 1.



**Figure 1.** Profiles of species-specific stocking densities for three treatments.

<sup>1</sup> Profile Analysis is usually found in the section covering “repeated measures” in multivariate statistics textbooks such as Johnson and Wichern (1994), Morrison (1990), or Tabachnick and Fidell (1989).

<sup>2</sup> The motivation for this example was a study conducted by George Harper. It is slightly unusual because rather than measuring tree density at several distinct time periods, we have measured density for distinct species at one time.

The objectives of the experiment are to test the following null hypotheses:

$H_0$ : The pattern of the four species' densities is consistent for all levels of the brushing treatment (parallel profiles).

$H_0$ : The density for each species is not affected by the brushing treatment (coincident profiles).

$H_0$ : There is no difference between the four species' densities within each treatment level (horizontal profiles).

The design is a one-way completely randomized design with subsampling that involves the observation of multiple responses (i.e. the different species' densities). It is *not* a two-way factorial experiment because species is merely an observed factor and not directly controlled (i.e. randomized) by the experimenter. The species' densities are more appropriately considered as part of the response instead of part of the design, especially since they will likely be dependent (i.e. the species will compete with one another) within each circular subplot. This interdependence violates a factorial ANOVA assumption of independent treatment response.

This situation could be viewed as a split-plot design by developing an *ad hoc* definition of the experimental unit for the split-plot factor species (e.g. all the trees of each species in a circular subplot). However, this approach involves an unreasonable covariance assumption<sup>3</sup> (Morrison, 1990), and an awkward adjustment factor for the  $F$  statistics' degrees of freedom (Looney and Stanley, 1989). Profile analysis is truly suited for this scenario.

Profile analysis is a specific style of multivariate analysis of variance (MANOVA). MANOVA is the natural choice for analyzing the type of data described above since it models the interdependencies among the response variables (e.g. different taxa). A drawback to MANOVA is that an observation having a missing value for any of the response variables is entirely excluded. Also, the power of the test statistic decreases as the number of the response variables increases.

Profile analysis addresses three questions:

1) Are the profiles **parallel**?

Two other questions, contingent on the answer being “yes” to question (1) are:

2) Are the profiles **coincident** (i.e. identical or superimposed)?

3) Are the profiles **level** (i.e. horizontal or flat)? This test is somewhat contingent on the answer to question (2), but more on this later.

Questions (2) and (3) are logical only if the profiles are parallel. If they are not parallel, one must proceed by either (a) testing the treatment effect separately for each species using ANOVA or (b) assessing the species effect separately for each treatment group using MANOVA. Remember to control the overall type I error rate when conducting multiple tests (e.g. Bonferroni correction).

In profile analysis, the different responses must be expressed on the same scale, and the experimental units between and within treatment groups must be independent of one another. In the example, this was assured by randomly assigning the treatments to the square plots. It is also important that the number of response variables  $p$  is less than or equal to the number of experimental units in the smallest treatment group. MANOVA assumes that the  $p$ -dimensional responses from

---

<sup>3</sup> The assumption, termed sphericity, requires that the covariance structure between the responses for all possible pairs of response variables are equal. For our example, this would imply that for each subplot, the covariance between larch and fir equals the covariance between larch and pine ... equals the covariance between pine and birch.

each treatment group ( $i = 1, 2, \dots, k$ ) belong to multivariate normal populations with unique mean vectors  $\bar{\mu}_i$  that share a common, though unknown, covariance matrix  $\Sigma$ . A simple and usually sufficient way to assess this assumption is to examine the residuals (from the fitted model) separately for each response<sup>4</sup>. The upcoming sections will address the three null hypotheses for the brushing example, which auspiciously correspond to the questions answered by profile analysis. The resultant SAS<sup>®</sup> output (SAS Institute, Inc., 1990) can be found in Appendix 2.

## 1. Test of Parallel Profiles

Testing the first null hypothesis is the same as the test of parallel profiles, or of no species  $\times$  treatment interaction. This test is achieved by working with the successive differences in the responses. The profiles are parallel if the differences are the same across the three treatment groups. For  $p$  response variables, the  $(p - 1)$  differences will be dependent. Hence, MANOVA is used to test if the group of  $(p - 1)$  differences is the same for all treatment levels.

There are several different multivariate test statistics available for the test of parallel profiles, all of which will generally yield equivalent results. The four most common are: *Wilk's Lambda*, *Pillai's Trace*, the *Hotelling-Lawley Trace*, and *Roy's Greatest Root*. For most examples of profile analysis, *Wilk's Lambda* is desirable because it can be converted exactly to an  $F$  statistic. The form of the relationship between *Wilk's Lambda* and  $F$  depends on both the number of treatment levels and the number of response variables. Because this relationship can be quite complicated, the reader is referred to pg. 248 of Johnson and Wichern (1992) for specific details. In instances such as ours where  $(p - 1) \geq 1$  and the number of treatment levels equals 3, *Wilk's Lambda* can be converted exactly to an  $F$  statistic distributed with  $2(p - 1)$  and  $2(N - (p - 1) - 2)$  degrees of freedom. Here,  $N = 15$  is the total number of experimental units. The required SAS<sup>®</sup> code to test for parallel profiles looks something like:

```
proc glm data=herb;
  title "Test for Parallel Profiles";
  class trt rep;
  model larch fir pine birch = trt rep(trt) / nouni;
  manova h=trt e=rep(trt) m=larch-fir,fir-pine,pine-birch prefix=diff;
run;
```

There are some notable features in the above code:

1. In the `model` statement, all of the  $p$  response variables must be specified on the left-hand side of the “=”.
2. The `nouni` option may be used to suppress any univariate output.
3. The `manova` statement<sup>5</sup> must be used to specify the correct hypothesis ( $h=<\text{treatment effect}>$ ) and error ( $e=<\text{error term}>$ ) terms, which is analogous to the `test` statement in univariate analyses. Specifying the error term is necessary because of the subsampling in the design.

<sup>4</sup> More elaborate methods are available to assess multivariate normality. See for example, Mardia (1974) or contact the author of this pamphlet for a diagnostic SAS<sup>®</sup> program.

<sup>5</sup> When the design is simple and balanced, all three profile analysis tests are available in SAS<sup>®</sup> using the `repeated` statement within `proc glm`. However, the user cannot specify explicit error terms in conjunction with this statement, and the test for level profiles will be incorrect for seriously unbalanced designs,

4. The `m` option must be used to specify the correct so-called ‘contrast’ or ‘transformation’ matrix. In this case, successive differences are to be used<sup>6</sup>. This option is unnecessary if subtractions are performed earlier, say during the data preparation stage.
5. The `prefix` option may be used to label the differences as ‘diff1’, ‘diff2’, ..., ‘diff( $p-1$ )’.

The resultant test statistics, using data from the example are: *Wilk’s Lambda* = 0.53 or equivalently  $F_{6,20} = 1.25$  with  $p\text{-value} = 0.32$  (Appendix 2). Thus, the null hypothesis of parallel profiles is not rejected, and we may proceed with the other two tests.

## 2. Test of Coincident Profiles

The second null hypothesis investigates whether the profiles are coincident or superimposed on one another. Since the profiles are parallel, one only needs to examine the average (or total) of the  $p$  responses for each treatment level. Hence, the test is univariate.

The calculation of the average (or total) response for each group can be done during the data preparation stage, and then followed by a standard univariate analysis. However, since the data have already been arranged in a multivariate configuration for the first test, an easier approach is to use the `m` option in conjunction with a `manova` statement; a `manova` statement is still required because a multivariate grouping is being averaged (or totaled). Suitable SAS<sup>®</sup> code would look something like:

```
proc glm data=herb;
  title "Test for Coincident Profiles (given that they're parallel)";
  class trt rep;
  model larch fir pine birch = trt rep(trt) / nouni;
  manova h=trt e=rep(trt) m=larch+fir+pine+birch prefix=total;
run;
```

The resultant degrees of freedom for the  $F$ -test of this completely randomized design with subsampling are  $k-1=2$  and  $N-k=12$  (where  $k$  is the number of treatment levels, and  $N$  is the total number of experimental units). Using the example data, we obtain  $F_{2,12} = 2.76$  with  $p\text{-value} = 0.10$  (Appendix 2). Therefore, we do not reject the hypothesis of coincident profiles. In order to truly *accept* the null hypothesis of coincident profiles, the power of the test should be computed. See Bergerud (1995) for a discussion of this point.

## 3. Test of Level Profiles

The final null hypothesis of equal mean density among species can be tested by examining if the profile(s) is level. This purely multivariate test requires only that the test of parallelism has not been rejected. However, the results of the previous test also affect the approach here. This is because we can test either (a) if each of the profiles are level, or (b) if a single combined profile, pooled over the three treatment groups, is level.

A simple way to approach this problem is to again take successive differences<sup>7</sup>. If the profile(s) is indeed level, then the  $(p-1)$  differences should each be *zero*. The SAS<sup>®</sup> code to test the pooled profile is:

---

<sup>6</sup> Interestingly, the results of multivariate tests are not affected by the choice for the  $(p-1) \times p$  contrast matrix, so long as its rows sum to zero. The relevance of this technical result is that the manner in which the four species are ordered does not matter.

<sup>7</sup> An alternative approach (found in many textbooks) is to subtract the overall grand mean from each of the  $p$  response variables, and then test whether these new variables are each zero. This yields equivalent results to our method.

```
proc glm data=herb;
  title "Test for Level Profiles (given that they're parallel and coincident)";
  class trt rep;
  model larch fir pine birch = rep(trt) / nouni;
  manova h=intercept e=rep(trt) m=larch-fir,fir-pine,pine-birch prefix=diff / htype=1;
run;
```

A manova statement is required to declare that the (multivariate) intercept is being tested against zero ( $h=intercept$ ). The stipulation of the error term ( $e=<error\ term>$ ) is again needed because of the subsampling, and the  $m$  option is required to compute successive differences. Note that the model statement does not include the treatment effect. Further, the explicit specification of the Type I hypothesis matrix<sup>8</sup> ( $htype=1$ ) is especially crucial for unbalanced designs.

As before, *Wilk's Lambda* will be used as the statistic to test the null hypothesis (but there are also others) because it can be converted to an exact  $F$  statistic with  $p-1$  and  $N-(p-1)$  degrees of freedom. For our example  $F_{3,12} = 19.50$  (*Wilk's Lambda* = 0.17) has  $p\text{-value} = 0.00007$  (Appendix 2). Therefore, we reject the null hypothesis of equal density levels across species.

If the test of coincidence had been rejected, then the full model must be specified<sup>9</sup>. For example, had the test for coincidence been rejected, the SAS<sup>®</sup> code would now look like:

```
proc glm data=herb;
  title "Test for Level Profiles (given that they're parallel but not coincident)";
  class trt rep;
  model larch fir pine birch = trt rep(trt) / nouni;
  manova h=intercept e=rep(trt) m=larch-fir,fir-pine,pine-birch prefix=diff / htype=1;
run;
```

This approach will produce smaller elements in the error matrix leading to a larger observed  $F$ -statistic. However, this is somewhat offset by a reduction in the denominator degrees of freedom for the  $F$ -statistic because more parameters must be estimated. The degrees of freedom are now  $p-1$  and  $N-(p-1)-(k-1)$ . Using this model, the results of the example are

$F_{3,10} = 20.00$  (*Wilk's Lambda* = 0.14) with  $p\text{-value} = 0.00015$  (Appendix 2). Therefore, we still would reject the null hypothesis of equal density levels across species.

In conclusion, the researcher has shown that while the densities are different between the four species, the relationships between the densities are unaffected by the treatment.

### Prepared by:

Peter Ott. Phone: 250-387-7982; e-mail: Peter.Ott@gems7.gov.bc.ca  
B.C. Ministry of Forests, Research Branch.

<sup>8</sup> This is analogous to choosing the Type I (sequential) sums of squares in univariate ANOVA. This method is required because the sums of squares (and cross products) for the intercept term should not be conditioned on the inclusion of any other terms in the model.

<sup>9</sup> This will appropriately partition the variability due to the treatment effect.

## References:

- Bergerud, W.A. 1995. Post-hoc power analyses for ANOVA F-tests. Biom. Info. Pamph. 52. Res. Br. B.C. Min. For., Victoria, B.C.
- Johnson, R.A. and D.W. Wichern. 1992. Applied Multivariate Statistical Analysis, Third Edition. Prentice-Hall, Inc. New Jersey.
- Looney, S.W. and W.B. Stanley. 1989. Exploratory repeated measures analysis for two or more groups. The American Statistician. 43(4):220-225.
- Mardia, K.V. 1974. Applications of some measures of multivariate skewness and kurtosis in testing normality and robustness studies. Sankhya B. 36:115-128.
- Morrison, D.F. 1990. Multivariate Statistical Methods, Third Edition. McGraw-Hill. New York.
- Tabachnick, B.G. and L.S. Fidell. 1989. Using Multivariate Statistics, Second Edition. Harper Collins Publishers, Inc. New York.
- SAS Institute, Inc. 1990. SAS/STAT User's Guide, Version 6, Fourth Edition, Volume 2. SAS Institute, Inc. Cary, NC. pp. 891-996.

## Appendix 1: Data

Treatment	Replicate	Subsample	Density (stems per hectare)			
			Larch	Fir	Pine	Birch
herbicide	1	1	17.89	54.77	73.98	59.80
herbicide	1	2	56.15	72.93	62.24	50.05
herbicide	1	3	31.60	52.15	49.89	19.73
herbicide	2	1	47.47	43.82	72.95	51.03
herbicide	2	2	13.38	32.81	102.48	90.66
herbicide	2	3	4.31	39.03	68.64	67.29
herbicide	3	1	41.03	43.32	86.47	55.16
herbicide	3	2	23.24	50.69	72.03	51.10
herbicide	3	3	13.87	46.26	86.10	53.55
herbicide	4	1	4.40	54.13	102.61	64.67
herbicide	4	2	17.97	38.35	59.64	72.32
herbicide	4	3	0.69	42.89	103.03	79.92
herbicide	5	1	60.32	54.47	50.49	59.78
herbicide	5	2	46.02	60.54	65.60	65.44
herbicide	5	3	41.40	33.08	72.33	84.85
manual	1	1	20.89	49.77	68.98	64.80
manual	1	2	59.15	67.93	57.24	55.05
manual	1	3	34.60	47.15	44.89	24.73
manual	2	1	50.47	38.82	67.95	56.03
manual	2	2	16.38	27.81	97.48	95.66
manual	2	3	7.31	34.03	63.64	72.29
manual	3	1	44.03	38.32	81.47	60.16
manual	3	2	26.24	45.69	67.03	56.10
manual	3	3	16.87	41.26	81.10	58.55
manual	4	1	7.40	49.13	97.61	69.67
manual	4	2	20.97	33.35	54.64	77.32
manual	4	3	3.69	37.89	98.03	84.92
manual	5	1	63.32	49.47	45.49	64.78
manual	5	2	49.02	55.54	60.60	70.44
manual	5	3	44.40	28.08	67.33	89.85
control	1	1	20.8	19.84	95.51	85.58
control	1	2	26.42	53.31	104.49	84.17
control	1	3	24.55	50.56	74.28	70.34
control	2	1	0.00	25.04	93.73	91.52
control	2	2	8.82	6.49	95.15	90.94
control	2	3	29.25	32.10	87.88	103.19
control	3	1	35.06	34.94	79.20	80.29
control	3	2	19.66	40.86	70.78	72.78
control	3	3	23.93	52.68	61.63	62.06
control	4	1	14.19	30.71	106.59	102.44
control	4	2	42.76	46.69	79.61	71.61
control	4	3	37.99	49.96	57.66	66.73
control	5	1	31.52	54.11	75.38	70.83
control	5	2	27.02	28.59	74.43	83.07
control	5	3	0.66	11.24	101.92	118.91

## Appendix 2: Edited SAS® Output

### ----- Test for Parallel Profiles -----

Manova Test Criteria and F Approximations for  
the Hypothesis of no Overall TRT Effect  
on the variables defined by the M Matrix Transformation  
H = Type III SS&CP Matrix for TRT    E = Type III SS&CP Matrix for REP(TRT)

Statistic	Value	F	Num DF	Den DF	Pr > F
Wilks' Lambda	0.52867810	1.2511	6	20	.32318
Pillai's Trace	0.50808985	1.2487	6	22	.32052
Hotelling-Lawley Trace	0.82196320	1.2329	6	18	.33561
Roy's Greatest Root	0.72619410	2.6627	3	11	.09980

NOTE: F Statistic for Roy's Greatest Root is an upper bound.

NOTE: F Statistic for Wilks' Lambda is exact.

### ----- Test for Coincident Profiles (given that they're parallel) -----

Manova Test Criteria and Exact F Statistics for  
the Hypothesis of no Overall TRT Effect  
on the variables defined by the M Matrix Transformation  
H = Type III SS&CP Matrix for TRT    E = Type III SS&CP Matrix for REP(TRT)

Statistic	Value	F	Num DF	Den DF	Pr > F
Wilks' Lambda	0.68497490	2.7594	2	12	.10329
Pillai's Trace	0.31502510	2.7594	2	12	.10329
Hotelling-Lawley Trace	0.45990752	2.7594	2	12	.10329
Roy's Greatest Root	0.45990752	2.7594	2	12	.10329

### ----- Test for Level Profiles (given that they're parallel and coincident) -----

Manova Test Criteria and Exact F Statistics for  
the Hypothesis of no Overall INTERCEPT Effect  
on the variables defined by the M Matrix Transformation  
H = Type I SS&CP Matrix for INTERCEPT    E = Type III SS&CP Matrix for REP(TRT)

Statistic	Value	F	Num DF	Den DF	Pr > F
Wilks' Lambda	0.17022981	19.4976	3	12	7.E-05
Pillai's Trace	0.82977019	19.4976	3	12	7.E-05
Hotelling-Lawley Trace	4.87441169	19.4976	3	12	7.E-05
Roy's Greatest Root	4.87441169	19.4976	3	12	7.E-05

### ----- Test for Level Profiles (given that they're parallel but not coincident) -----

Manova Test Criteria and Exact F Statistics for  
the Hypothesis of no Overall INTERCEPT Effect  
on the variables defined by the M Matrix Transformation  
H = Type I SS&CP Matrix for INTERCEPT    E = Type III SS&CP Matrix for REP(TRT)

Statistic	Value	F	Num DF	Den DF	Pr > F
Wilks' Lambda	0.14287545	19.9970	3	10	.00015
Pillai's Trace	0.85712455	19.9970	3	10	.00015
Hotelling-Lawley Trace	5.99910290	19.9970	3	10	.00015
Roy's Greatest Root	5.99910290	19.9970	3	10	.00015